Problem:
Score : 500 points

Problem Statement

There are N towns on a plane. The i-th town is located at the coordinates (x_i,y_i). There may be more than one town at the same coordinates.

You can build a road between two towns at coordinates (a,b) and (c,d) for a cost of min(|a-c|,|b-d|) yen (the currency of Japan). It is not possible to build other types of roads.

Your objective is to build roads so that it will be possible to travel between every pair of towns by traversing roads. At least how much money is necessary to achieve this?

Constraints

2 ≤ N ≤ 10^5

0 ≤ x_i,y_i ≤ 10^9

All input values are integers.

Input

Input is given from Standard Input in the following format:

N
x_1 y_1
x_2 y_2
:
x_N y_N

Output

Print the minimum necessary amount of money in order to build roads so that it will be possible to travel between every pair of towns by traversing roads.

Sample Input 1

3
1 5
3 9
7 8

Sample Output 1

3

Build a road between Towns 1 and 2, and another between Towns 2 and 3. The total cost is 2+1=3 yen.

Sample Input 2

6
8 3
4 9
12 19
18 1
13 5
7 6

Sample Output 2

8

Solution in OCaml (4.02.3) (judge verdict: Accepted)

Scanf.scanf "%d" (fun n ->
    let xy = Array.init n (fun i -> Scanf.scanf " %d %d" (fun x y -> x, y, i)) in
    Array.sort compare xy;
    let module S = Set.Make (struct type t = int * int * int let compare = compare end) in
    let rec loop j acc =
        if j = n then acc else
            let px, _, pi = xy.(j - 1) in
            let cx, _, ci = xy.(j)     in
            let acc = S.add (cx - px, pi, ci) acc in
            loop (j + 1) acc
    in
    let edges = loop 1 S.empty in
    Array.sort (fun (x0, y0, _) (x1, y1, _) -> compare (y0, x0) (y1, x1)) xy;
    let rec loop j acc =
        if j = n then acc else
            let _, py, pi = xy.(j - 1) in
            let _, cy, ci = xy.(j)     in
            let acc = S.add (cy - py, pi, ci) acc in
            loop (j + 1) acc
    in
    let edges = loop 1 edges in

    let par = Array.init n (fun i -> i) in
    let rec root x =
        if x = par.(x) then x else
            let r = root par.(x) in
            let () = par.(x) <- r in
            r
    in

    let merge a b =
        if root a <> root b then par.(par.(b)) <- par.(a)
    in

    let rec loop edges cost =
        if S.is_empty edges then cost else (
            let ((c, n1, n2) as elt) = S.min_elt edges in
            let edges = S.remove elt edges in
            if root n1 = root n2 then loop edges cost else (
                merge n1 n2;
                loop edges (cost + c)
            )
        )
    in
    loop edges 0 |> Printf.printf "%d\n"
)